
Abstract. The paper “Scrap your boilerplate” (SYB) introduces a com-
binator library for generic programming that offers generic traversals and
queries. Classically, support for generic programming consists of two es-
sential ingredients: a way to write (type-)overloaded functions, and in-
dependently, a way to access the structure of data types. SYB seems to
lack the second. As a consequence, it is difficult to compare with other
approaches such as PolyP or Generic Haskell. In this paper we reveal
the structural view that SYB builds upon. This allows us to define the
combinators as generic functions in the classical sense. We explain the
SYB approach in this changed setting from ground up, and use the un-
derstanding gained to relate it to other generic programming approaches.
Furthermore, we show that the SYB view is applicable to a very large
class of data types, including generalized algebraic data types.

1 Introduction

The paper “Scrap your boilerplate” (SYB) [1] introduces a combinator library
for generic programming that offers generic traversals and queries. Classically,
support for generic programming consists of two essential ingredients: a way
to write (type-)overloaded functions, and independently, a way to access the
structure of data types. SYB seems to lacks the second, because it is entirely
based on combinators.

In this paper, we make the following contributions:

– We explain the SYB approach from ground up using an explicit representa-
tion of data types, the spine view. Many of the SYB library functions are
more easily defined in the spine view than using the combinators underlying
the original SYB library.

– We compare the expressive power and applicability of the spine view to the
original SYB paper, to PolyP [2] and to Generic Haskell [3, 4].

– Furthermore, we show that the SYB view is applicable to a very large class
of data types, including generalized algebraic data types (GADTs) [5, 6].



2 R. Hinze, A. Löh, and B. Oliveira

We use Haskell [7] for all our examples. The source code of this paper [8]
constitutes a Haskell program that can be compiled by GHC [9] in order to
test and experiment with our implementation. While our implementation is not
directly usable as a separate library, because it is not extensible (new data types
cannot be added in a compositional way), this deficiency is not tied to the idea
of the Spine view: We show in Section 8 how to integrate the Spine view with
the third SYB paper, thereby providing an extensible, albeit slightly less elegant
implementation.

In this introduction, we explain the ingredients of a system for generic pro-
gramming, and argue that the original SYB presentation does not clearly qualify
as such a system. In order to better understand the concept of generic program-
ming, let us first look at plain functional programming.

1.1 Functional programming and views

As functional programmers in a statically typed language, we are used to define
functions by case analysis on a data type. In fact, it is standard practice to define
a function on a data type by performing case analysis on the input. The shape
of the data type guides our function definitions, and affects how easy it is to
define certain functions.

As an example, assume we want to implement a priority queue supporting
among others the operation

splitMinimum :: PriorityQueue → Maybe (Int ,PriorityQueue)

to separate the minimum from the remaining queue if the queue is not empty. We
can choose a heap-structured tree to implement the priority queue, and define

data Tree a = Empty | Node (Tree a) a (Tree a)
type PriorityQueue = Tree Int .

The choice of a heap as the underlying data stucture makes the implementation
of splitMinimum slightly tricky, requiring an auxiliary operation to merge two
heaps.

If, on the other hand, we choose a sorted list to represent the priority queue

data PriorityQueue = Void | Min Int PriorityQueue ,

we make our life much easier, because splitMinimum is now trivial to define.
The price we pay is that the implementation on lists is likely to be less efficient
than the one using the tree. Such different views on a data structure need not
be mutually exclusive. Wadler and others have proposed language support for
views [10, 11].

Many functions on a single data type follow common traversal and recursion
patterns. Instead of defining each function by case analysis, it is possible to
define combinators that capture these patterns. For instance, given functions



“Scrap Your Boilerplate” Reloaded 3

foldTree :: r → (r → a → r → r) → Tree a → r
mapTree :: (a → b) → Tree a → Tree b ,

we can write functions to perform an inorder traversal of the tree or to increase
every label in a tree by one very concisely:

inorder = foldTree [ ] (λl x r → l ++ [x ] ++ r)
incTree = mapTree (+1) .

1.2 Generic programming

A generic function is a function that is defined once, but works for many data
types. It can adapt itself to the structure of data types. Generic functions are
also called polytypic or structurally polymorphic.

Genericity is different from parametric polymorphism, where the same code
works for multiple types, and the structure of a data type is not available for
analysis. It is also more specific than ad-hoc polymorphism, which allows a
function to be defined for different data types, by providing one implementation
for each type.

Typical examples of generic functions are equality or comparison, parsing and
unparsing, serialization, traversals over large data structures and many others.

Support for generic programming consists of two essential ingredients. Firstly,
support for ad-hoc polymorphism is required. This allows the programmer to
write overloaded functions, i.e., functions that dispatch on a type argument.
Secondly, we need a generic view on the structure of data types. In a nominal
type system, types with similar structure are considered to be completely dis-
tinct. To employ generic programming, we need to lower this barrier and make
the structure transparent if desired.

The two ingredients are orthogonal, and for both, there is a choice. Over-
loaded functions can be expressed in Haskell using the class system, using a
type-safe cast operation, by reflecting the type system on the value level, or
by a combination of the above. Any of these approaches has certain advantages
and disadvantages, but they are mostly interchangeable and do not dramatically
affect the expressivity of the generic programming system.

The structural view, on the other hand, dictates the flavour of the whole
system: it affects the set of data types we can represent in the view, the class of
functions we can write using case analysis on the structure, and potentially the
efficiency of these functions. The structural view is used to make an overloaded
function truly generic, working for a data type even if it has no ad-hoc case for
that type.

For instance, PolyP views data types as fixed points of regular functors.
Therefore its approach is limited to regular data types, but the view allows access
to the points of recursion and allows the definition of recursion combinators such
as catamorphisms. Generic Haskell uses a sum-of-products view which is more
widely applicable, but limits the class of functions we can write. The concept



4 R. Hinze, A. Löh, and B. Oliveira

of generic views is explained further in a recent paper [12], and is related to
universes in dependently-typed programming [13].

In summary, it turns out that there is a close analogy between plain functional
and generic programming: the concepts of views, function definition by case
analysis, and combinators occur in both settings.

1.3 Scrap your boilerplate

In analogy with the situation on plain functions, not all generic functions are
defined by case analysis. Just as there are powerful combinators for ordinary
functions, such combinators also exist for generic programming. In fact, the very
combinators we have used above, foldTree and mapTree, are typical candidates
for generalization.

The paper “Scrap your boilerplate” (SYB) describes a library for strategic
programming [14], i.e., it offers combinators for generic traversals and queries on
terms. Two central combinators of the SYB library are everywhere to traverse a
data structure and modify it in certain places, and everything to traverse a data
structure and collect information in the process.

The SYB approach builds completely on combinators, and some fundamental
combinators are assumed to be provided by the implementation. While this is
fine in practice, it makes it difficult to compare SYB with other approaches
such as PolyP or Generic Haskell. The reason is that the concept of a generic
view seems to be missing. Functions are never defined by case analysis on the
structure of types.

However, the generic view is only hidden in the original presentation. In this
paper we reveal the structure that SYB uses behind the scenes and that allows
us to define the SYB combinators as generic functions by case analysis on that
structure.

We will explain the SYB approach in this changed setting from ground up.
The focus of the presentation is on conceptual conciseness. We do not strive to
replace the original implementation, but to complement it by an alternative im-
plementation which may be easier to understand and relate to other approaches.

1.4 Organization of this paper

The rest of this paper is organized as follows: We first describe the two orthogonal
ingredients required for generic programming in our presentation of the SYB
approach: overloaded functions (Section 2) and the spine view, the structure
that is the hidden foundation of SYB (Section 3). We then review the central
combinators of SYB in Section 4. Section 5 shows how we can access names of
constructors.

In Section 6, we take a step back and relate SYB to other generic program-
ming approaches. Inspired by our analysis on the expressiveness of the SYB
approach, we demonstrate how to extend the spine view to generalized algebraic
data types (Section 7).

In Section 8, we take a more detailed look at the way that extensible generic
functions are encoded in the third of the SYB papers. We explain how the spine
view can be used with a class-based encoding of overloaded functions.

Section 9 discusses related work and concludes.

2 Overloaded functions

The standard way in Haskell to express an overloaded function is to use a type
class. In fact, this is the way taken by the original SYB papers: in the first
SYB paper, type classes are used in conjunction with a type-safe cast operation,
and in the third paper, overloaded functions are expressed solely based on type
classes. However, type classes leave it to the compiler to find the correct instance,
and thus hide a non-trivial aspect of the program. In this paper, we prefer to
be more explicit and emphasize the idea that an overloaded function dispatches
on a type argument. Haskell excels at embedded languages, so it seems a good
idea to try to embed the type language in Haskell. The following way to encode
overloaded functions is not new: it is based on Hinze’s “Fun of Programming”
chapter [15] and has been used widely elsewhere [16].

The whole point of static types is that they can be used at compile time to
distinguish programs, hence we certainly do not want to use an unparameter-
ized data type Type to represent types. Instead, we add a parameter so that
Type t comprises only type representations for the type t . We now need ways
to construct values of type Type t . For instance, Int can be a representation of
the type Int , so that we have Int :: Type Int . Similarly, if we have a representa-
tion r of type a, we can make List r a representation of type [a ], or formally
List :: Type a → Type [a ].

The notation we use suggests that Int and List are data constructors of type
Type, but this impossible in Haskell 98, because the result type of a constructor
must always be unrestricted, i.e., Type a for some type variable a. Fortunately,
GHC now supports generalized algebraic data types (GADTs) [5, 6], which lift
exactly this restriction. Therefore, we can indeed define Type in Haskell using
the following GADT:

data Type :: ∗ → ∗ where
Int :: Type Int
Char :: Type Char
List :: Type a → Type [a ]
Pair :: Type a → Type b → Type (a, b)
Tree :: Type a → Type (Tree a) .

This type allows us to represent integers, characters, lists, pairs, and trees –
enough to give an example of a simple overloaded function that sums up all
integers in a value:

sum :: Type a → a → Int
sum Int n = n

sum Char = 0
sum (List a) xs = foldr (+) 0 (map (sum a) xs)
sum (Pair a b) (x , y) = sum a x + sum b y
sum (Tree a) t = sum (List a) (inorder t) .

The function sum works on all types that can be constructed from Int , Char ,
[ ], (,), and Tree, for instance, on a complex type such as [(Char , Int)]: the
expression

sum (List (Pair Char Int)) [(’k’, 6), (’s’, 9), (’ ’, 27)]

evaluates to 42.
The function sum is an example of an ad-hoc-polymorphic function. There

are a limited number of cases for different types, defining potentially unrelated
behavior of sum for these types. The function will not work on types such as
Bool or Maybe or even on a type

newtype MyPair a b = MyPair (a, b) ,

because Haskell has a nominal type system, hence MyPair a b is isomorphic to,
yet distinct from (a, b).

3 The spine view

In this section, we learn how to define a truly generic sum, which works on Bool
and Maybe and MyType, among others.

Take a look at any Haskell value. If it is not of some abstract type, it can
always be written as a data constructor applied to other values. For example,
Node Empty 2 Empty is the Node data constructor, applied to the three values
Empty , 2, and Empty . Even built-in types such as Int or Char are not funda-
mentally different: every literal can be seen as a nullary constructor.

Let us make the structure of constructed values visible and mark each con-
structor using Constr , and each function application using ♦. The example from
above becomes

Constr Node ♦ Empty ♦ 2 ♦ Empty .

The functions Constr and (♦)3 are themselves constructors of a new data type
Spine:4

3 We use (♦) as a symbol for an infix data constructor. For our presentation, we ignore
the Haskell rule that names of infix data constructors must start with a colon.

4 Note that in contrast to Type, the data type Spine is not necessarily a generalized
algebraic data type. The result types of the constructors are not restricted, Spine
could therefore be defined in GHC as a normal data type with existentials. However,
we prefer the GADT syntax.

data Spine :: ∗ → ∗ where
Constr :: a → Spine a
(♦) :: Spine (a → b) → a → Spine b .

Given a value of type Spine a, we can recover the original value of type a by
undoing the conversion step made before:

fromSpine :: Spine a → a
fromSpine (Constr c) = c
fromSpine (f ♦ a) = (fromSpine f ) a .

The function fromSpine is parametrically polymorphic, i.e., it works indepen-
dently of the type in question: it just replaces Constr with the original construc-
tor and (♦) with function application.

Unfortunately, fromSpine is the only interesting function we can write on a
Spine. Reconsider the type of the (♦) constructor:

(♦) :: Spine (a → b) → a → Spine b .

The type a is not visible in the final result (it is existentially quantified in the
data type), so the only thing we can do with the component of type a is to
combine it somehow with the component of type Spine (a → b).

Since we intend to call overloaded functions on the value of type a, we require
a representation of the type of a. Our solution is thus that together with the
value of type a, we store a representation of its type. To this end, we introduce
a data type for typed values5

data Typed a = a : Type a ,

and then adapt (♦) to use Typed a instead of a:

data Spine :: ∗ → ∗ where
Constr :: a → Spine a
(♦) :: Spine (a → b) → Typed a → Spine b .

Of course, we have to adapt fromSpine to ignore the new type annotations:

fromSpine :: Spine a → a
fromSpine (Constr c) = c
fromSpine (f ♦ (a : )) = (fromSpine f ) a .

We can define a right inverse to fromSpine, as an overloaded function. For each
data type, the definition follows a trivial pattern. Here are the cases for the Int
and Tree types:

toSpine :: Type a → a → Spine a
toSpine Int n = Constr n

5 We use (:) as data constructor for the type Typed in this paper. The cons-operator
for lists, written (:) in Haskell, does not occur in this paper.

toSpine (Tree a) Empty = Constr Empty
toSpine (Tree a) (Node l x r) = Constr Node

♦ (l : Tree a) ♦ (x : a) ♦ (r : Tree a) .

With all the machinery in place, we can now write the truly generic sum:

sum :: Type a → a → Int
sum Int n = n
sum t x = sum ′ (toSpine t x )
sum ′ :: Spine a → Int
sum ′ (Constr c) = 0
sum ′ (f ♦ (x : t)) = sum ′ f + sum t x .

This function requires only a single type-specific case, namely the one for Int .
The reason is that we want to do something specific for integers, which does not
follow the general pattern, whereas the formerly explicit behavior for the types
Char , [ ], (,), and Tree is now completely subsumed by the function sum ′. Note
also that in the last line of sum ′, the type information t for x is indispensable,
as we call the generic function sum recursively.

Why are we in a better situation than before? If we encounter a new data
type such as Maybe, we still have to extend the representation type

data Type :: ∗ → ∗ where
. . .
Maybe :: Type a → Type (Maybe a)

and adapt the toSpine function

fromMaybe :: Type a → Maybe a → Spine (Maybe a)
fromMaybe Nothing = Constr Nothing
fromMaybe a (Just x ) = Constr Just ♦ (x : a)
toSpine :: Type a → a → Spine a
toSpine . . . = . . .
toSpine (Maybe a) = fromMaybe a .

However, this has to be done only once per data type, and it is so simple that it
could easily be done automatically. The code for the generic functions (of which
there can be many) is completely unaffected by the addition of a new data type.

4 Generic queries and traversals

In this section, we implement the two central SYB combinators everything and
everywhere that are used to construct generic queries and traversals.

4.1 Generic queries

A query is an overloaded function that returns a result of a specific type:

type Query r = ∀a.Type a → a → r .

We have already seen an example of a query, namely the sum function from
Section 3. There are many more applications of queries: computation of the size
of a structure, collection of names, collection of free variables, building a finite
map, finding a specific element etc.

If we look back at the generic sum function, we see that it performs several
tasks at once, leading to a relatively complex definition: integers are preserved,
while in general, constructors are replaced by 0; the subresults are added; finally,
a recursive traversal is performed over the entire data structure.

In the following, we describe how to separate these different activities into
different functions, and at the same time abstract from the specific problem of
summing up values.

If we already have a query, we can define a derived query that applies the
original query to all immediate children of a given constructor:

mapQ :: Query r → Query [r ]
mapQ q t = mapQ ′ q ◦ toSpine t
mapQ ′ :: Query r → (∀a.Spine a → [r ])
mapQ ′ q (Constr c) = [ ]
mapQ ′ q (f ♦ (x : t)) = mapQ ′ q f ++ [q t x ] .

The results of the original query q are collected in a list. The combinator mapQ
does not traverse the input data structure. The traversal is the job of everything ′,
which is defined in terms of mapQ :

everything ′ :: Query r → Query [r ]
everything ′ q t x = [q t x ] ++ concat (mapQ (everything ′ q) t x ) .

Here, we apply the given query q to the entire argument x , and then recurse for
the immediate children. The SYB version of everything fuses everything ′ with
an application of foldl1 , using a binary operator to combine all the elements of
the nonempty list returned by everything ′:

everything :: (r → r → r) → Query r → Query r
everything op q t x = foldl1 op ([q t x ] ++ mapQ (everything op q) t x ) .

In order to express the query sum in terms of everything , we need a simple query
sumQ expressing that we want to count integers:

sumQ :: Query Int
sumQ Int n = n
sumQ t x = 0
sum :: Query Int
sum = everything (+) sumQ .

4.2 Generic traversals

While a query computes an answer of a fixed type from an input, a traversal is
an overloaded function that preserves the type of its input:

type Traversal = ∀a.Type a → a → a .

The counterpart of mapQ is mapT . It applies a given traversal to the immediate
children of a constructor, then rebuilds a value of the same constructor from the
results:

mapT :: Traversal → Traversal
mapT h t = fromSpine ◦mapT ′ h ◦ toSpine t
mapT ′ :: Traversal → (∀a.Spine a → Spine a)
mapT ′ h (Constr c) = Constr c
mapT ′ h (f ♦ (x : t)) = mapT ′ h f ♦ (h t x : t) .

The function mapT not only consumes a value of the type argument, but also
produces one. Therefore we call not only toSpine on the input value, but also
fromSpine before returning the result. The calls to fromSpine and toSpine are
determined by the type of the generic function that is defined. The general
principle is described elsewhere [3, Chapter 11].

Using mapT , we can build bottom-up or top-down variants of everywhere,
which apply the given traversal recursively:

everywherebu :: Traversal → Traversal
everywherebu f t = f t ◦mapT (everywherebu f ) t
everywheretd :: Traversal → Traversal
everywheretd f t = mapT (everywheretd f ) t ◦ f t .

There are many applications of traversals, such as renaming variables in an
abstract syntax tree, annotating a structure with additional information, opti-
mizing or simplifying a structure etc. Here is a simplified example of a transfor-
mation performed by the Haskell refactorer HaRe [17], which rewrites a Haskell
if construct into an equivalent case expression according to the rule

if e then e1 else e2 Ã case e of True → e1;False → e2 .

We assume a suitable abstract syntax for Haskell. The rewrite rule is captured
by the traversal

ifToCaseT :: Traversal
ifToCaseT HsExp (HsIf e e1 e2) =

HsCase e [HsAlt (HsPLit (HsBool True)) e1,
HsAlt (HsPLit (HsBool False)) e2 ]

ifToCaseT e = e .

The traversal can be applied to a complete Haskell program using

ifToCase = everywherebu ifToCaseT .

5 Generically showing values

We have seen that we can traverse data types in several ways, performing po-
tentially complex calculations in the process. However, we cannot reimplement
Haskell’s show function, even though it looks like a Query String . The reason
is that there is no way to access the name of a constructor. We have a case for
constructors, Constr , in our Spine data type, but there is really not much we
can do at this point. So far, we have either invented a constant value ([ ] in the
case of mapQ), or applied the constructor itself again (in the case of mapT ).

But it is easy to provide additional information for each constructor. When
we define toSpine for a specific data type, whether manually or automatically,
we have information about the constructors of the data type available, so why
not use it? Let us therefore modify Spine once more:

data Spine :: ∗ → ∗ where
As :: a → ConDescr → Spine a
(♦) :: Spine (a → b) → Typed a → Spine b .

We have renamed Constr to As, as we intend to use it as a binary operator
which takes a constructor function and information about the constructor. In
this paper, we use only the name to describe a constructor,

type ConDescr = String ,

but we could include additional information such as its arity, the name of the
type, the “house number” of the constructor and so on. Adapting Spine means
that the generation of toSpine has to be modified as well. We show as an example
how to do this for the type Tree:

toSpine (Tree a) Empty = Empty ‘As‘ "Empty"
toSpine (Tree a) (Node l x r) = Node ‘As‘ "Node"

♦ (l : Tree a) ♦ (x : a) ♦ (r : Tree a) .

With the new version of Spine, the function show is straightforward to write:

show :: Type a → a → String
show t x = show ′ (toSpine t x )
show ′ :: Spine a → String
show ′ ( ‘As‘ c) = c
show ′ (f ♦ (a : t)) = "(" ++ show ′ f ++ " " ++ show t a ++ ")" .

The result of the call

show (Tree Int) (Node (Node Empty 1 Empty) 2 (Node Empty 3 Empty))

is the string

“(((Node (((Node Empty) 1) Empty)) 2) (((Node Empty) 3) Empty))” .

It is also easy to define toConDescr , which returns the ConDescr for a value:

toConDescr :: Type a → a → ConDescr
toConDescr t = toConDescr ′ ◦ toSpine t
toConDescr ′ :: Spine a → String
toConDescr ′ ( ‘As‘ c) = c
toConDescr ′ (f ♦ (a : t)) = toConDescr ′ f .

Even though we have information about constructors, we cannot define a
generic read without further extensions. In the next section, we will discuss this
and other questions regarding the expressivity of the SYB approach.

6 SYB in context

In the previous sections, we have introduced the SYB approach on the basis of
the Spine data type. Generic functions are overloaded functions that make use
of the Spine view by calling toSpine on their type argument.

We have seen that we can define useful and widely applicable combinators
such as everything and everywhere using some basic generic functions. As long
as we stay within the expressivity of these combinators, it is possible to perform
generic programming avoiding explicit case analysis on types.

In this section, we want to answer how expressive the Spine view is in com-
parison to both the original presentation of SYB, which uses only a given set
of combinators, and in relation to other views, as they are employed by other
approaches to generic programming such as PolyP and Generic Haskell.

6.1 The original presentation

As described in the section of implementing SYB in the original paper, it turns
out that mapT and mapQ are both instances of a function that is called gfoldl .
We can define gfoldl , too. To do this, let us define the ordinary fold (or cata-
morphism, if you like) of the Spine type:

foldSpine :: (∀a.a → r a) → (∀a b.r (a → b) → Typed a → r b) →
Spine a → r a

foldSpine constr (¨) (c ‘As‘ ) = constr c
foldSpine constr (¨) (f ♦ (x : t)) = (foldSpine constr (¨) f ) ¨ (x : t) .

The definition follows the catamorphic principle of replacing data constructors
with functions. The SYB gfoldl is just foldSpine composed with toSpine:

gfoldl :: Type a → (∀a.a → r a) → (∀a b.r (a → b) → Typed a → r b) →
a → r a

gfoldl t constr app = foldSpine constr app ◦ toSpine t .

It is therefore clear that our approach via the Spine type and the original SYB
approach via gfoldl are in principle equally expressive, because the Spine type
can be recovered from gfoldl .

However, we believe that the presence of the explicit data type Spine makes
the definitions of some generic functions easier, especially if they do not directly
fall in the range of any of the simpler combinators.

The original SYB paper describes only generic functions that either consume
a value based on its type (queries, consumers), or that consume a value based
on its type and build up a similar value at the same time (traversals). There are
also generic functions that construct values based on a type (producers). Such
functions include the already mentioned generic read , used to parse a string into
a value of a data type, or some, a function that produces some non-bottom value
of a given data type. We cannot define such functions without further help: The
definition of some would presumably follow the general pattern of overloaded
functions on spines, the shape of the final case dictated by the type of some
(cf. Section 4.2):

some :: Type a → a
some . . . = . . .
some t = fromSpine some ′

But we cannot define some ′ ::Spine a, because that would yield fromSpine some ′

:: ∀a.a, which has to be ⊥ according to the parametricity theorem [18]. Due to
the well-definedness of fromSpine, some ′ would have to be ⊥, too.

It is nevertheless possible to define some :: Type a → a, but only if Type is
augmented with more information about the type it represents. In particular,
it must be possible to obtain a full list of constructors in some suitable data
structure from a type representation. If we pursue this path further, we end up
with a specific data type to hold the constructors of a single type

data TypeSpine :: ∗ → ∗ where
TypeAs :: a → ConDescr → TypeSpine a
(@) :: TypeSpine (a → b) → Type a → TypeSpine b

This data type is almost the same as Spine, only the second argument of (@) is
a Type instead of a Typed . A TypeSpine therefore contains no value except the
constructor function itself. We can now write a function

constrs :: Type a → [TypeSpine a ]

for each data type, but this function has to be defined for each data type sepa-
rately, much like toSpine.

Here is an example for the constructors of Tree:

constrs (Tree a) = [Empty ‘TypeAs‘ "Empty",
Node ‘TypeAs‘ "Node" @ Tree a @ a @ Tree a ] .

With constrs, we now can define some or even read :

some :: Type a → a
some = some ′ ◦ head ◦ constrs

some ′ :: TypeSpine a → a
some ′ (TypeAs c ) = c
some ′ (f @ a) = some ′ f (some a) .

Indeed, the gunfold function that is added to the set of predefined combinators
in the second SYB paper [19] is derived from the catamorphism on Constr much
like gfoldl is derived from the catamorphism on Spine. Unfortunately though,
there is no relation between gunfold and an anamorphism on Spine.

We can, however, define functions on multiple type arguments without further
additions. The definition of generic equality is very straightforward using the
spine view:

eq :: Type a → Type b → a → b → Bool
eq t1 t2 x y = eq ′ (toSpine t1 x ) (toSpine t2 y)
eq ′ :: Spine a → Spine b → Bool
eq ′ ( ‘As‘ c1) ( ‘As‘ c2) = c1 = = c2

eq ′ (f1 ♦ (a1 : t1)) (f2 ♦ (a2 : t2)) = eq ′ f1 f2 ∧ eq t1 t2 a1 a2

eq ′ = False .

The generalized type of eq avoids the necessity of a type-level equality test. In
the second SYB paper, eq is defined in terms of a combinator called zipWithQ .

type Query2 r = ∀a b.Type a → Type b → a → b → r
zipWithQ :: Query2 r → Query2 [r ]
zipWithQ q t1 t2 x y = zipWithQ ′ q (toSpine t1 x ) (toSpine t2 y)
zipWithQ ′ :: Query2 r → (∀a b.Spine a → Spine b → [r ])
zipWithQ ′ q (f1 ♦ (a1 : t1)) (f2 ♦ (a2 : t2)) = zipWithQ ′ q f1 f2 ++

[q t1 t2 a1 a2 ]
zipWithQ ′ q s1 s2 = [ ]
eqQ :: Query2 Bool
eqQ Int Int x y = x = = y
eqQ Char Char x y = x = = y
eqQ t1 t2 x y = toConDescr t1 x = = toConDescr t2 y
eq :: Query2 Bool
eq t1 t2 x y = and ([eqQ t1 t2 x y ] ++ zipWithQ eq t1 t2 x y) .

Although we can mirror the definition of zipWithQ , we believe that the direct
definition is much clearer.

6.2 Other views and their strengths and weaknesses

Let us now look at two other approaches to generic programming, PolyP and
Generic Haskell. They are also based on overloaded functions, but they do not
represent values using Spine. A different choice of view affects the class of generic
functions that can be written, how easily they can be expressed, and the data
types that can be represented.

PolyP In PolyP [2], data types of kind ∗ → ∗ are viewed as fixed points of
regular pattern functors. The regular functors in turn are of kind ∗ → ∗ → ∗
and represented as lifted sums of products. The view makes use of the following
type definitions:

data Fix f = In (f (Fix f ))
type LSum f g a r = Either (f a r) (g a r)
type LProd f g a r = (f a r , g a r)
type LUnit a r = ()
type Par a r = a
type Rec a r = r .

Here, Fix is a fixed-point computation on the type level. The type constructors
LSum, LProd , and LUnit are lifted variants of the binary sum type Either , the
binary product type (,), and the unit type (). Finally we have Par to select the
parameter, and Rec to select the recursive call.

As an example, our type Tree has pattern functor TreeF :

data TreeF a r = EmptyF | NodeF r a r .

We have (modulo ⊥) that Tree a ∼= Fix (TreeF a). Furthermore, we can view
TreeF as a binary sum (it has two constructors), where the right component is
a nested binary product (NodeF has three fields). The recursive argument r is
represented by Rec, the parameter to Tree by Par :

type TreeFS a r = LSum LUnit (LProd Rec (LProd Par Rec)) a r .

Again, we have (modulo ⊥) an isomorphism TreeF a r ∼= TreeFS a r .
The view of PolyP has two obvious disadvantages: first, due to its two-level

nature, it is relatively complicated; second, it is quite limited in its applicability.
Only data types of kind ∗ → ∗ that are regular can be represented.

On the other hand, many generic functions on data types of kind ∗ → ∗ are
definable. PolyP can express functions to parse, compare, unify, or print values
generically. Its particular strength is that recursion patterns such as cata- or
anamorphisms can be expressed generically, because each data type is viewed as
a fixed point, and the points of recursion are visible.

Generic Haskell In contrast to PolyP, Generic Haskell [3, 4] uses a view that
is much more widely applicable and is slightly easier to handle: all data types
are (unlifted) sums of products. The data type Tree is viewed as the isomorphic

type TreeS a = Either () (Tree a, (a,Tree a)) .

The original type Tree appears in TreeS , there is no special mechanism to treat
recursion differently. This has a clear advantage, namely that the view is appli-
cable to nested and mutually recursive data types of arbitrary kinds. In fact, in
Generic Haskell all Haskell 98 data types can be represented. The price is that
recursion patterns such as cata- or anamorphisms cannot be defined directly.

On the other hand, generic functions in Generic Haskell can be defined such
that they work on types of all kinds. It is therefore significantly more powerful
than PolyP. In Generic Haskell we can, for instance, define a generic map that
works for generalized rose trees, a data type of kind (∗ → ∗) → ∗ → ∗:

data Rose f a = Fork a (f (Rose f a)) .

Scrap your boilerplate The Spine view is not so much based on the structure
of types, but on the structure of values. It emphasizes the structure of a construc-
tor application. We have already noticed that this limits the generic functions
that can be written. Pure producers such as read or some require additional
information. Furthermore, all generic functions work on types of kind ∗. It is not
possible to define a generic version of map for type constructors, or to define a
recursion pattern such as a catamorphism generically.

But the Spine view also has two noticeable advantages over the other views
discussed. Firstly, the view is simple, and the relation between a value and
its spine representation is very direct. As a consequence, the transformation
functions fromSpine and toSpine are quite efficient, and it is easy to deforest the
Spine data structure.

Secondly, as every (non-abstract) Haskell value is a constructor application,
the view is very widely applicable. Not only all Haskell 98 data types of all
kinds can be represented, the Spine view is general enough to represent data
types containing existentials and even GADTs without any further modifica-
tions. This is particularly remarkable, because at the moment, GHC does not
support automatic derivation of classes for GADTs. The methods of the classes
Eq , Ord , and Show can easily be defined using the SYB approach. Thus, there
is no theoretical problem to allow derivation of these classes also for GADTs.
We discuss this newly found expressive power further in the next section.

7 Scrap your boilerplate for “Scrap your boilerplate”

There are almost no limits to the data types we can represent using Spine. One
very interesting example is the GADT of types itself, namely Type. This allows
us to instantiate generic functions on type Type. Consequently, we can show
types by invoking the generic show function, or compute type equality using the
generic equality eq ! Both are useful in the context of dynamically typed values:

data Dynamic where
Dyn :: t → Type t → Dynamic .

The difference between the types Dynamic and Typed is that Dynamic contains
an existential quantification.

Before we can actually use generic functions on Dynamic, we require that
Type has a constructor for types and dynamic values:

data Type :: ∗ → ∗ where
. . .

Type :: Type a → Type (Type a)
Dynamic :: Type Dynamic .

The function toSpine also requires cases for Type and Dynamic, but converting
types or dynamics into the Spine view is entirely straightforward, as the following
example cases demonstrate:

toSpine (Type a ′) (Type a) = Type ‘As‘ "Type" ♦ (a : Type a)
toSpine Dynamic (Dyn x t) = Dyn ‘As‘ "Dyn" ♦ (x : t) ♦ (t : Type t) .

In the first line above, a ′ is always equal to a, but the Haskell type system does
not know that, so we do not enforce it in the program. The output of

show Dynamic (Dyn (Node Empty 2 Empty) (Tree Int))

is now the string

“((Dyn (((Node Empty) 2) Empty)) (Tree Int))” ,

and comparing the dynamic value to itself using eq Dynamic Dynamic yields
indeed True, incorporating a run-time type equality test.

8 Class-based implementation

In this section, we combine the idea of the Spine view with an alternative way
to express overloaded functions, using classes as described in the third SYB
paper [20], short SYB 3 in the remainder of this section. We assume knowledge
with the material in that paper, as some of the techniques described there are
quite subtle.

The advantage of this presentation (which comes at the price of elegance)
is that the resulting functions are extensible in a compositional way. New data
types can be added, and functions adapted, just by providing a few additional
instances.

As an introduction, here is how the overloaded sum function from Section 2
is expressed using type classes:

class Sum a where
sum :: a → Int

instance Sum Int where
sum n = n

instance Sum Char where
sum = 0

instance Sum a ⇒ Sum [a ] where
sum xs = foldr (+) 0 (map sum xs)

instance (Sum a,Sum b) ⇒ Sum (a, b) where
sum (x , y) = sum x + sum y

instance Sum a ⇒ Sum (Tree a) where
sum t = sum (inorder t) .

If we want to turn this into the generic variant of sum, as given near the end of
Section 3, while keeping the function extensible in the spirit of SYB 3, we have
to do a bit of preliminary work.

The whole point of the exercise is to reuse the spine view, but to do without
explicit type representations. It turns out that the main thing we need to be
able to do is to turn values into their spine representation, so we define a Data
class as follows:

class (Sat (ctx a)) ⇒ Data ctx a where
toSpine :: a → Spine ctx a ,

which plays more or less the same role as the Data class in the SYB papers.
The intricacies of Sat and the ctx parameter are explained in SYB 3, they

have to do with tying the recursive knot for extensible generic functions. The
definition of Sat (read: satisfy) is simply

class Sat a where
dict :: a .

We can now specify the variant of the Spine data type we use here, where we
use the Data class instead of the Type data type:

data Spine :: (∗ → ∗) → (∗ → ∗) where
As :: a → ConDescr → Spine ctx a
(♦) :: (Data ctx a) ⇒ Spine ctx (a → b) → a → Spine ctx b .

Defining instances of the Data class is without surprises:

instance Sat (ctx Int) ⇒ Data ctx Int where
toSpine n = n ‘As‘ Prelude.show n

instance (Sat (ctx (Tree a)),Data ctx a) ⇒ Data ctx (Tree a) where
toSpine Empty = Empty ‘As‘ "Empty"
toSpine (Node l x r) = Node ‘As‘ "Node" ♦ l ♦ x ♦ r .

Now we can start defining the generic sum function. The main part of the func-
tion turns out to be simpler than in the non-generic case, because we need only
one specific case, for integers:

class Sum a where
sum :: a → Int

instance Sum Int where
sum n = n .

The generic case looks as follows:

instance Data SumD a ⇒ Sum a where
sum x = sum ′ (toSpine x ) .

The data type SumD is a dictionary for the sum function. We use it for recursive
calls:

data SumD a = SumD{sumD :: a → Int }
sum ′ :: Spine SumD a → Int
sum ′ (c ‘As‘ ) = 0
sum ′ (f ♦ x ) = sum ′ f + sumD dict x
instance Sum a ⇒ Sat (SumD a) where

dict = SumD{sumD = sum } .

This function works as the original generic sum, only that we do not have to
provide a type argument. The call

sum (Node Empty (17 :: Int) (Node Empty 25 Empty))

evaluates to 42. The function sum is easy to extend: if we want to handle new
data types, all we have to do is to define appropriate instances for the Data and
Sum classes.

When moving to the classic SYB combinators, we have to resort to proxies
as explained in SYB 3:

data Proxy (ctx :: ∗ → ∗) -- empty type
proxy :: Proxy ctx
proxy = error "proxy"

type Query ctx r = ∀a.Data ctx a ⇒ Proxy ctx → a → r
mapQ :: Query ctx r → Query ctx [r ]
mapQ q = mapQ ′ q ◦ toSpine
mapQ ′ :: Query ctx r → (∀a.Spine ctx a → [r ])
mapQ ′ q (c ‘As‘ ) = [ ]
mapQ ′ q (f ♦ x ) = mapQ ′ q f ++ [q proxy x ]
everything ′ :: Query ctx r → Query ctx [r ]
everything ′ q p x = [q p x ] ++ concat (mapQ (everything ′ q) p x )
everything :: (r → r → r) → Query ctx r → Query ctx r
everything op q p x = foldl1 op ([q p x ] ++ mapQ (everything op q) p x ) .

With these preparations in place, we can now write sum as a query:

class SumQ a where
sumQ :: a → Int
sumQ = 0

instance SumQ Int where
sumQ n = n

data SumQD a = SumQD{sumQD :: a → Int }
instance SumQ a ⇒ Sat (SumQD a) where

dict = SumQD{sumQD = sumQ }
sum :: Query SumQD Int
sum = everything (+) (const (sumQD dict)) .

The rest of the code in this paper can be adapted to a class-based approach
using the same techniques as shown above. This demonstrates that the choice of
how to represent overloaded functions is mostly independent of the usage of the
spine view.

9 Conclusions

The SYB approach has been developed by Peyton Jones and Lämmel in a series
of papers [1, 19, 20]. Originally, it was an implementation of strategic program-
ming [14] in Haskell, intended for traversing and querying complex, compound
data such as abstract syntax trees.

The ideas underlying the generic programming extension PolyP [2] go back to
the categorical notions of functors and catamorphisms, which are independent of
the data type in question [21]. Generic Haskell [22] was motivated by the desire
to overcome the restrictions of PolyP.

Due to the different backgrounds, it is not surprising that SYB and generic
programming have remained difficult to compare for a long time. The recent work
on generic views [12, 23] has been an attempt to unify different approaches. We
believe that we bridged the gap in this paper for the first time, by presenting
the Spine data type which encodes the SYB approach faithfully.

Our implementation handles the two central ingredients of generic program-
ming differently from the original SYB paper: we use overloaded functions with
explicit type arguments instead of overloaded functions based on a type-safe
cast [1] or a class-based extensible scheme [20]; and we use the explicit spine
view rather than a combinator-based approach. Both changes are independent
of each other, and have been made with clarity in mind: we think that the struc-
ture of the SYB approach is more visible in our setting, and that the relations
to PolyP and Generic Haskell become clearer. We have revealed that while the
spine view is limited in the class of generic functions that can be written, it is
applicable to a very large class of data types, including GADTs.

Our approach cannot be used easily as a library, because the encoding of
overloaded functions using explicit type arguments requires the extensibility of
the Type data type and of functions such as toSpine. One can, however, incor-
porate Spine into the SYB library while still using the techniques of the SYB
papers to encode overloaded functions.

In this paper, we do not use classes at all, and we therefore expect that it
is easier to prove algebraic properties about SYB (such as mapT copy = copy
where copy = id is the identity traversal) in this setting. For example, we
believe that the work of Reig [24] could be recast using our approach, leading to
shorter and more concise proofs.

4. Löh, A., Jeuring, J., Clarke, D., Hinze, R., Rodriguez, A., de Wit, J.: The Generic
Haskell user’s guide, version 1.42 (Coral). Technical Report UU-CS-2005-004, In-
stitute of Information and Computing Sciences, Utrecht University (2005)

5. Xi, H., Chen, C., Chen, G.: Guarded recursive datatype constructors. In: Proceed-
ings of the ACM SIGPLAN-SIGACT symposium on Principles of Programming
Languages (POPL 2003), ACM Press (2003) 224–235

6. Peyton Jones, S., Washburn, G., Weirich, S.: Wobbly types: Type inference for
generalised algebraic data types. Technical Report MS-CIS-05-26, University of
Pennsylvania (2004)

7. Peyton Jones, S., ed.: Haskell 98 Language and Libraries: The Revised Report.
Cambridge University Press (2003)

